contract NyxToken is Context, IERC20, Ownable {
    using Address for address;

    mapping (address => uint256) private _rOwned;
    mapping (address => uint256) private _tOwned;
    mapping (address => mapping (address => uint256)) private _allowances;

    mapping (address => bool) private _isExcludedFromFee;
    mapping (address => bool) private _isExcluded;
    address[] private _excluded;

    // Liquidity Pairs
    mapping (address => bool) public _isPair;

    // Banned contracts
    mapping (address => bool) public _isBanned;
 
    uint256 private constant MAX = ~uint256(0);
    uint256 public constant _tTotal = 1e15 * 10**9;
    uint256 private _rTotal = (MAX - (MAX % _tTotal));
    uint256 private _tFeeTotal;

    string private constant _name = "Nyx Token";
    string private constant _symbol = "NYXT";
    uint8 private constant _decimals = 9;

    // Wallets
    address public _devWalletAddress;   // General charity wallet - 1%
    address public _msWalletAddress;    // Multiple Sclerosis wallet - 2%
    address public _vetWalletAddress;   // Veterans wallet - 2%
    address public _liqWalletAddress;   // liquidity injection - 2%
    address public _deadAddress = 0x000000000000000000000000000000000000dEaD;

    // Fees - all fee values represent 0.1%

    // Buy Fees
    uint256 public _buyTaxFee = 20; // 2%
    uint256 public _buyDevFee = 10; // 1%
    uint256 public _buyLiquidityFee = 20;   // 2%

    // Sell Fees
    uint256 public _sellTaxFee = 20;
    uint256 public _sellDevFee = 10;
    uint256 public _sellLiquidityFee = 20;

    // Fees (Current)
    uint256 private _taxFee;
    uint256 private _devFee;
    uint256 private _liquidityFee;

    bool public _contractFeesEnabled = true;

    IUniswapV2Router02 public uniswapV2Router;
    address public uniswapV2Pair;
    
    bool private inSwapAndLiquify;
    bool public swapAndLiquifyEnabled = true;
    bool public swapEnabled = true;
    
    uint256 public _maxTxAmount = _tTotal;
    uint256 public numTokensSellToAddToLiquidity = 1e12 * 10**9; // 0.001%
    
    // Events
    event SwapAndLiquifyEnabledUpdated(bool enabled);
    event SwapAndLiquify(
        uint256 tokensSwapped,
        uint256 ethReceived,
        uint256 tokensIntoLiquidity
    );
    event NumTokensSellToAddToLiquidityUpdated(uint256 numTokensSellToAddToLiquidity);
    event SetContractFeesEnabled(bool _bool);
    event RouterSet(address _router);
    event SetIsPair(address _address, bool _bool);
    event SetIsBanned(address _address, bool _bool);
    event SetSwapEnabled(bool enabled);
    event SetDevWalletAddress(address _address);
    event SetLiqWalletAddress(address _address);
    event WithdrawalBNB(uint256 _amount, address to);
    event WithdrawalToken(address _tokenAddr, uint256 _amount, address to);
    
    modifier lockTheSwap {
        inSwapAndLiquify = true;
        _;
        inSwapAndLiquify = false;
    }
    
    constructor(address router, address devWallet, address msWallet, address vetWallet) {
        _rOwned[owner()] = _rTotal;
        
        _setRouter(router);
        _devWalletAddress = devWallet;
        _msWalletAddress = msWallet;
        _vetWalletAddress = vetWallet;
        
        // Exclude owner, dev wallet, liq wallet, and this contract from fee
        _isExcludedFromFee[owner()] = true;
        _isExcludedFromFee[_devWalletAddress] = true;
        _isExcludedFromFee[_msWalletAddress] = true;
        _isExcludedFromFee[_vetWalletAddress] = true;
        _isExcludedFromFee[address(this)] = true;
        
        emit Transfer(address(0), owner(), _tTotal);
    }

    function name() external pure returns (string memory) {
        return _name;
    }

    function symbol() external pure returns (string memory) {
        return _symbol;
    }

    function decimals() external pure returns (uint8) {
        return _decimals;
    }

    function totalSupply() external pure override returns (uint256) {
        return _tTotal;
    }

    function balanceOf(address account) public view override returns (uint256) {
        if (_isExcluded[account]) return _tOwned[account];
        return tokenFromReflection(_rOwned[account]);
    }

    function transfer(address recipient, uint256 amount) external override returns (bool) {
        _transfer(_msgSender(), recipient, amount);
        return true;
    }

    function allowance(address owner, address spender) external view override returns (uint256) {
        return _allowances[owner][spender];
    }

    function approve(address spender, uint256 amount) external override returns (bool) {
        _approve(_msgSender(), spender, amount);
        return true;
    }

    function transferFrom(address sender, address recipient, uint256 amount) external override returns (bool) {
        _transfer(sender, recipient, amount);
        _approve(sender, _msgSender(), _allowances[sender][_msgSender()] - amount);
        return true;
    }

    function increaseAllowance(address spender, uint256 addedValue) external virtual returns (bool) {
        _approve(_msgSender(), spender, _allowances[_msgSender()][spender] + addedValue);
        return true;
    }

    function decreaseAllowance(address spender, uint256 subtractedValue) external virtual returns (bool) {
        _approve(_msgSender(), spender, _allowances[_msgSender()][spender] - subtractedValue);
        return true;
    }

    function isExcludedFromReward(address account) external view returns (bool) {
        return _isExcluded[account];
    }

    function totalFees() external view returns (uint256) {
        return _tFeeTotal;
    }

    function reflectionFromToken(uint256 tAmount, bool deductTransferFee) external view returns(uint256) {
        require(tAmount <= _tTotal, "Amount must be less than supply");
        if (!deductTransferFee) {
            (uint256 rAmount,,,,,,) = _getValues(tAmount);
            return rAmount;
        } else {
            (,uint256 rTransferAmount,,,,,) = _getValues(tAmount);
            return rTransferAmount;
        }
    }

    function tokenFromReflection(uint256 rAmount) public view returns(uint256) {
        require(rAmount <= _rTotal, "Amount must be less than total reflections");
        uint256 currentRate =  _getRate();
        return rAmount / (currentRate);
    }

    function excludeFromReward(address account) public onlyOwner() {
        require(!_isExcluded[account], "Account is already excluded");
        if(_rOwned[account] > 0) {
            _tOwned[account] = tokenFromReflection(_rOwned[account]);
        }
        _isExcluded[account] = true;
        _excluded.push(account);
    }

    function includeInReward(address account) external onlyOwner() {
        require(_isExcluded[account], "Account is already included");
        for (uint256 i = 0; i < _excluded.length; i++) {
            if (_excluded[i] == account) {
                _excluded[i] = _excluded[_excluded.length - 1];
                _tOwned[account] = 0;
                _isExcluded[account] = false;
                _excluded.pop();
                break;
            }
        }
    }
    
    function _transferBothExcluded(address sender, address recipient, uint256 tAmount) private {
        (uint256 rAmount, uint256 rTransferAmount, uint256 rFee, uint256 tTransferAmount, uint256 tFee, uint256 tLiquidity, uint256 tdev) = _getValues(tAmount);
        _tOwned[sender] = _tOwned[sender] - (tAmount);
        _rOwned[sender] = _rOwned[sender] - (rAmount);
        _tOwned[recipient] = _tOwned[recipient] + (tTransferAmount);
        _rOwned[recipient] = _rOwned[recipient] + (rTransferAmount);        
        _takeLiquidity(tLiquidity);
        _takeMS(tFee);
        _takeVet(tFee);
        _takedev(tdev);
        _takeBurn(tdev);
        _reflectFee(rFee, tFee);
        emit Transfer(sender, recipient, tTransferAmount);
    }
    
    function excludeFromFee(address account) external onlyOwner {
        _isExcludedFromFee[account] = true;
    }
    
    function includeInFee(address account) external onlyOwner {
        _isExcludedFromFee[account] = false;
    }
    
    function setBuyTaxFeePercent(uint256 taxFee) external onlyOwner() {
        _buyTaxFee = taxFee;
        require(_buyTaxFee + _buyDevFee + _buyLiquidityFee <= 100, "Total fees exceed 10%");
    }

    function setBuyDevFeePercent(uint256 devFee) external onlyOwner() {
        _buyDevFee = devFee;
        require(_buyTaxFee + _buyDevFee + _buyLiquidityFee <= 100, "Total fees exceed 10%");
    }
    
    function setBuyLiquidityFeePercent(uint256 liquidityFee) external onlyOwner() {
        _buyLiquidityFee = liquidityFee;
        require(_buyTaxFee + _buyDevFee + _buyLiquidityFee <= 100, "Total fees exceed 10%");
    }

    function setSellTaxFeePercent(uint256 taxFee) external onlyOwner() {
        _sellTaxFee = taxFee;
        require(_sellTaxFee + _sellDevFee + _sellLiquidityFee <= 100, "Total fees exceed 10%");
    }

    function setSellDevFeePercent(uint256 devFee) external onlyOwner() {
        _sellDevFee = devFee;
        require(_sellTaxFee + _sellDevFee + _sellLiquidityFee <= 100, "Total fees exceed 10%");
    }
    
    function setSellLiquidityFeePercent(uint256 liquidityFee) external onlyOwner() {
        _sellLiquidityFee = liquidityFee;
        require(_sellTaxFee + _sellDevFee + _sellLiquidityFee <= 100, "Total fees exceed 10%");
    }
   
    function setMaxTxPercent(uint256 maxTxPercent) external onlyOwner() {
        _maxTxAmount = _tTotal * (maxTxPercent) / (
            10**2
        );
    }

    function setSwapAndLiquifyEnabled(bool _enabled) external onlyOwner() {
        swapAndLiquifyEnabled = _enabled;
        emit SwapAndLiquifyEnabledUpdated(_enabled);
    }
    
    function setSwapEnabled(bool _enabled) external onlyOwner() {
        swapEnabled = _enabled;
        emit SetSwapEnabled(_enabled);
    }
    
    function setDevWalletAddress(address _address) external onlyOwner() {
        require(_address != address(0), "Error: devWallet address cannot be zero address");
        _devWalletAddress = _address;
        emit SetDevWalletAddress(_address);
    }

    function setMSWalletAddress(address _address) external onlyOwner() {
        require(_address != address(0), "Error: devWallet address cannot be zero address");
        _msWalletAddress = _address;
    }

    function setVetWalletAddress(address _address) external onlyOwner() {
        require(_address != address(0), "Error: devWallet address cannot be zero address");
        _vetWalletAddress = _address;
    }

    function setLiqWalletAddress(address _address) external onlyOwner() {
        require(_address != address(0), "Error: liqWallet address cannot be zero address");
        _liqWalletAddress = _address;
        emit SetLiqWalletAddress(_address);
    }
    
    function setNumTokensSellToAddToLiquidity(uint256 _amount) external onlyOwner() {
        require(_amount > 0);
        numTokensSellToAddToLiquidity = _amount;
        emit NumTokensSellToAddToLiquidityUpdated(_amount);
    }

    function setContractFeesEnabled(bool _bool) external onlyOwner() {
        _contractFeesEnabled = _bool;
        emit SetContractFeesEnabled(_bool);
    }
    
    function _setRouter(address _router) private {
        IUniswapV2Router02 _uniswapV2Router = IUniswapV2Router02(_router);
        uniswapV2Pair = IUniswapV2Factory(_uniswapV2Router.factory()).getPair(address(this), _uniswapV2Router.WETH());
        if(uniswapV2Pair == address(0))
            uniswapV2Pair = IUniswapV2Factory(_uniswapV2Router.factory()).createPair(address(this), _uniswapV2Router.WETH());
        uniswapV2Router = _uniswapV2Router;
        setIsPair(uniswapV2Pair, true);
        emit RouterSet(_router);
    }
    
    function setRouter(address _router) external onlyOwner() {
        _setRouter(_router);
    }
    
    // to receive ETH from uniswapV2Router when swapping
    receive() external payable {}

    // static reflection
    function _reflectFee(uint256 rFee, uint256 tFee) private {
        _rTotal = _rTotal - (rFee);
        _tFeeTotal = _tFeeTotal + (tFee);
    }

    function _getValues(uint256 tAmount) private view returns (uint256, uint256, uint256, uint256, uint256, uint256, uint256) {
        (uint256 tTransferAmount, uint256 tFee, uint256 tLiquidity, uint256 tdev) = _getTValues(tAmount);
        (uint256 rAmount, uint256 rTransferAmount, uint256 rFee) = _getRValues(tAmount, tFee, tLiquidity, tdev, _getRate());
        return (rAmount, rTransferAmount, rFee, tTransferAmount, tFee, tLiquidity, tdev);
    }

    function _getTValues(uint256 tAmount) private view returns (uint256, uint256, uint256, uint256) {
        uint256 tFee = calculateTaxFee(tAmount);    // 2%
        uint256 tLiquidity = calculateLiquidityFee(tAmount);
        uint256 tdev = calculateDevFee(tAmount);    // 1%
        // transfer amount = tAmount - MS fee (2%) - Vet fee (2%) - dev fee (1%) - liquidity fee (2%) - 1% burn
        uint256 tTransferAmount = tAmount - (2 * tFee) - (tLiquidity) - (2 * tdev);
        return (tTransferAmount, tFee, tLiquidity, tdev);
    }

    function _getRValues(uint256 tAmount, uint256 tFee, uint256 tLiquidity, uint256 tdev, uint256 currentRate) private pure returns (uint256, uint256, uint256) {
        uint256 rAmount = tAmount * (currentRate);
        uint256 rFee = tFee * (currentRate);
        uint256 rLiquidity = tLiquidity * (currentRate);
        uint256 rdev = tdev * (currentRate);
        uint256 rTransferAmount = rAmount - (2 * rFee) - (rLiquidity) - (2 * rdev);
        return (rAmount, rTransferAmount, rFee);
    }

    function _getRate() private view returns(uint256) {
        (uint256 rSupply, uint256 tSupply) = _getCurrentSupply();
        return rSupply / (tSupply);
    }

    function _getCurrentSupply() private view returns(uint256, uint256) {
        uint256 rSupply = _rTotal;
        uint256 tSupply = _tTotal;      
        for (uint256 i = 0; i < _excluded.length; i++) {
            if (_rOwned[_excluded[i]] > rSupply || _tOwned[_excluded[i]] > tSupply) return (_rTotal, _tTotal);
            rSupply = rSupply - (_rOwned[_excluded[i]]);
            tSupply = tSupply - (_tOwned[_excluded[i]]);
        }
        if (rSupply < _rTotal / (_tTotal)) return (_rTotal, _tTotal);
        return (rSupply, tSupply);
    }
    
    function _takeLiquidity(uint256 tLiquidity) private {
        uint256 currentRate =  _getRate();
        uint256 rLiquidity = tLiquidity * (currentRate);
        _rOwned[address(this)] = _rOwned[address(this)] + (rLiquidity);
        if(_isExcluded[address(this)])
            _tOwned[address(this)] = _tOwned[address(this)] + (tLiquidity);
    }
    // 1%
    function _takedev(uint256 tdev) private {
        uint256 currentRate =  _getRate();
        uint256 rdev = tdev * (currentRate);
        _rOwned[_devWalletAddress] = _rOwned[_devWalletAddress] + (rdev);
        if(_isExcluded[_devWalletAddress])
            _tOwned[_devWalletAddress] = _tOwned[_devWalletAddress] + (tdev);
    }

    function _takeBurn(uint256 tdev) private {
        uint256 currentRate =  _getRate();
        uint256 rdev = tdev * (currentRate);
        _rOwned[_deadAddress] = _rOwned[_deadAddress] + (rdev);
        if(_isExcluded[_deadAddress])
            _tOwned[_deadAddress] = _tOwned[_deadAddress] + (tdev);
    }

    // addition - take MS cut
    // 2%
    function _takeMS(uint256 tdev) private {
        uint256 currentRate =  _getRate();
        uint256 rdev = tdev * (currentRate);
        _rOwned[_msWalletAddress] = _rOwned[_msWalletAddress] + (rdev);
        if(_isExcluded[_msWalletAddress])
            _tOwned[_msWalletAddress] = _tOwned[_msWalletAddress] + (tdev);
    }

    // addition - take vet cut
    // 2%
    function _takeVet(uint256 tdev) private {
        uint256 currentRate =  _getRate();
        uint256 rdev = tdev * (currentRate);
        _rOwned[_vetWalletAddress] = _rOwned[_vetWalletAddress] + (rdev);
        if(_isExcluded[_vetWalletAddress])
            _tOwned[_vetWalletAddress] = _tOwned[_vetWalletAddress] + (tdev);
    }
    
    function calculateTaxFee(uint256 _amount) private view returns (uint256) {
        return _amount * (_taxFee) / (
            10**3
        );
    }

    function calculateDevFee(uint256 _amount) private view returns (uint256) {
        return _amount * (_devFee) / (
            10**3
        );
    }

    function calculateLiquidityFee(uint256 _amount) private view returns (uint256) {
        return _amount * (_liquidityFee) /(
            10**3
        );
    }
    
    function removeAllFee() private {
        _taxFee = 0;
        _devFee = 0;
        _liquidityFee = 0;
    }
    
    function isExcludedFromFee(address account) external view returns(bool) {
        return _isExcludedFromFee[account];
    }

    function _approve(address owner, address spender, uint256 amount) private {
        require(owner != address(0), "ERC20: approve from the zero address");
        require(spender != address(0), "ERC20: approve to the zero address");

        _allowances[owner][spender] = amount;
        emit Approval(owner, spender, amount);
    }

    function _transfer(
        address from,
        address to,
        uint256 amount
    ) private {
        require(from != address(0), "ERC20: transfer from the zero address");
        require(to != address(0), "ERC20: transfer to the zero address");
        require(amount > 0, "Transfer amount must be greater than zero");
        if(from != owner() && to != owner())
            require(amount <= _maxTxAmount, "Transfer amount exceeds the maxTxAmount.");
            
        if(!swapEnabled && (_isPair[to] || _isPair[from]))
            revert("Buying and selling is disabled");

        if(_isBanned[from] || _isBanned[to]) {
            revert("Address is banned");
        }

        // is the token balance of this contract address over the min number of
        // tokens that we need to initiate a swap + liquidity lock?
        // also, don't get caught in a circular liquidity event.
        // also, don't swap & liquify if sender is uniswap pair.
        uint256 contractTokenBalance = balanceOf(address(this));
        
        if(contractTokenBalance >= _maxTxAmount) {
            contractTokenBalance = _maxTxAmount;
        }
        
        bool overMinTokenBalance = contractTokenBalance >= numTokensSellToAddToLiquidity;
        if (
            overMinTokenBalance &&
            !inSwapAndLiquify &&
            _isPair[to] &&
            swapAndLiquifyEnabled &&
            !_isExcludedFromFee[from]
        ) {
            contractTokenBalance = numTokensSellToAddToLiquidity;
            swapAndLiquify(contractTokenBalance);
        }
        
        // Indicates if fee should be deducted from transfer
        bool takeFee = true;

        // Remove fees except for buying and selling
        if(!_isPair[from] && !_isPair[to]) {
            takeFee = false;
        }

        // Enable fees if contract fees are enabled and to or from is a contract
        if(_contractFeesEnabled && (from.isContract() || to.isContract())) {
            takeFee = true;
        }

        // If any account belongs to _isExcludedFromFee account then remove the fee
        if(_isExcludedFromFee[from] || _isExcludedFromFee[to]){
            takeFee = false;
        }

        // Set buy fees
        if(_isPair[from] || from.isContract()) {
            _taxFee = _buyTaxFee;
            _devFee = _buyDevFee;
            _liquidityFee = _buyLiquidityFee;
        }
        
        // Set sell fees
        if(_isPair[to] || to.isContract()) {
            _taxFee = _sellTaxFee;
            _devFee = _sellDevFee;
            _liquidityFee = _sellLiquidityFee;            
        }
        
        // Transfer amount, it will take tax, burn, liquidity fee
        _tokenTransfer(from,to,amount,takeFee);
    }

    function swapAndLiquify(uint256 contractTokenBalance) private lockTheSwap {
        // split the contract balance into halves
        uint256 half = contractTokenBalance / (2);
        uint256 otherHalf = contractTokenBalance - (half);

        // capture the contract's current ETH balance.
        // this is so that we can capture exactly the amount of ETH that the
        // swap creates, and not make the liquidity event include any ETH that
        // has been manually sent to the contract
        uint256 initialBalance = address(this).balance;

        // swap tokens for ETH
        swapTokensForEth(half); // <- this breaks the ETH -> HATE swap when swap+liquify is triggered

        // how much ETH did we just swap into?
        uint256 newBalance = address(this).balance - (initialBalance);

        // add liquidity to uniswap
        addLiquidity(otherHalf, newBalance);
        
        emit SwapAndLiquify(half, newBalance, otherHalf);
    }

    function swapTokensForEth(uint256 tokenAmount) private {
        // generate the uniswap pair path of token -> weth
        address[] memory path = new address[](2);
        path[0] = address(this);
        path[1] = uniswapV2Router.WETH();

        _approve(address(this), address(uniswapV2Router), tokenAmount);

        // make the swap
        uniswapV2Router.swapExactTokensForETHSupportingFeeOnTransferTokens(
            tokenAmount,
            0, // accept any amount of ETH
            path,
            address(this),
            block.timestamp
        );
    }

    function addLiquidity(uint256 tokenAmount, uint256 ethAmount) private {
        // approve token transfer to cover all possible scenarios
        _approve(address(this), address(uniswapV2Router), tokenAmount);

        // add the liquidity
        uniswapV2Router.addLiquidityETH{value: ethAmount}(
            address(this),
            tokenAmount,
            0, // slippage is unavoidable
            0, // slippage is unavoidable
            _liqWalletAddress,
            block.timestamp
        );
        emit Transfer(address(this), uniswapV2Pair, tokenAmount);
    }

    //this method is responsible for taking all fee, if takeFee is true
    function _tokenTransfer(address sender, address recipient, uint256 amount,bool takeFee) private {
        if(!takeFee)
            removeAllFee();
        
        if (_isExcluded[sender] && !_isExcluded[recipient]) {
            _transferFromExcluded(sender, recipient, amount);
        } else if (!_isExcluded[sender] && _isExcluded[recipient]) {
            _transferToExcluded(sender, recipient, amount);
        } else if (_isExcluded[sender] && _isExcluded[recipient]) {
            _transferBothExcluded(sender, recipient, amount);
        } else {
            _transferStandard(sender, recipient, amount);
        }
    }

    function _transferStandard(address sender, address recipient, uint256 tAmount) private {
        (uint256 rAmount, uint256 rTransferAmount, uint256 rFee, uint256 tTransferAmount, uint256 tFee, uint256 tLiquidity, uint256 tdev) = _getValues(tAmount);
        _rOwned[sender] = _rOwned[sender] - (rAmount);
        _rOwned[recipient] = _rOwned[recipient] + (rTransferAmount);
        _takeLiquidity(tLiquidity);
        _takedev(tdev);
        _takeBurn(tdev);
        _takeMS(tFee);
        _takeVet(tFee);
        _reflectFee(rFee, tFee);
        emit Transfer(sender, recipient, tTransferAmount);
    }

    function _transferToExcluded(address sender, address recipient, uint256 tAmount) private {
        (uint256 rAmount, uint256 rTransferAmount, uint256 rFee, uint256 tTransferAmount, uint256 tFee, uint256 tLiquidity, uint256 tdev) = _getValues(tAmount);
        _rOwned[sender] = _rOwned[sender] - (rAmount);
        _tOwned[recipient] = _tOwned[recipient] + (tTransferAmount);
        _rOwned[recipient] = _rOwned[recipient] + (rTransferAmount);           
        _takeLiquidity(tLiquidity);
        _takedev(tdev);
        _takeMS(tFee);
        _takeBurn(tdev);
        _takeVet(tFee);
        _reflectFee(rFee, tFee);
        emit Transfer(sender, recipient, tTransferAmount);
    }

    function _transferFromExcluded(address sender, address recipient, uint256 tAmount) private {
        (uint256 rAmount, uint256 rTransferAmount, uint256 rFee, uint256 tTransferAmount, uint256 tFee, uint256 tLiquidity, uint256 tdev) = _getValues(tAmount);
        _tOwned[sender] = _tOwned[sender] - (tAmount);
        _rOwned[sender] = _rOwned[sender] - (rAmount);
        _rOwned[recipient] = _rOwned[recipient] + (rTransferAmount);   
        _takeLiquidity(tLiquidity);
        _takedev(tdev);
        _takeMS(tFee);
        _takeBurn(tdev);
        _takeVet(tFee);
        _reflectFee(rFee, tFee);
        emit Transfer(sender, recipient, tTransferAmount);
    }

    function setIsPair(address _address, bool value) public onlyOwner() {
        _isPair[_address] = value;
        // add the pair for this token as the liquidity wallet for LP injections
        _liqWalletAddress = _address;
        emit SetIsPair(_address, value);
    }

    function setIsBanned(address _address, bool value) external onlyOwner() {
        require(_address.isContract(), "Error: Can only ban or unban contract addresses");
        _isBanned[_address] = value;
        emit SetIsBanned(_address, value);
    }

    function withdrawalToken(address _tokenAddr, uint _amount, address to) external onlyOwner() {
        IERC20 token = IERC20(_tokenAddr);
        token.transfer(to, _amount);
        emit WithdrawalToken(_tokenAddr, _amount, to);
    }
    
    function withdrawalBNB(uint _amount, address to) external onlyOwner() {
        require(address(this).balance >= _amount);
        payable(to).transfer(_amount);
        emit WithdrawalBNB(_amount, to);
    }

}
function _reflectFee(uint256 rFee, uint256 tFee) private {
    _rTotal = _rTotal - (rFee);
    _tFeeTotal = _tFeeTotal + (tFee);
}

// static reflection